

NetRexx

A JVM Language in the Clothes of Rexx

© 2011 Rony G. Flatscher (Rony.Flatscher@wu.ac.at)
Wirtschaftsuniversität Wien, Austria (http://www.wu.ac.at)

History, 1

- First non-Java language for the JVM
  - Much easier syntax than Java !
  - Programs use almost 1/3 less tokens than Java
- Authored by "father of Rexx"
  - IBM fellow Mike F. Cowlishaw
- Language specs published as a book
  - "The NetRexx Language" (TNRL), Cowlishaw M.F., Prentice Hall 1997
- Made available freely as "IBM EWS" ("IBM Employee Written Software)
  - However: "closed source"

History, 2

- Rexx Language Association (www.RexxLA.org)
  - Non-profit special interest group (SIG) for Rexx programmers
- Succeeded in convincing IBM to opensource Object REXX in 2004, first release by RexxLA in April 2005
  - *Very* successful project
- RexxLA NetRexx programmers interested in getting the source of NetRexx via RexxLA too
  - Opensourcing process succeeded on **2011-06-08**
- "RexxLA NetRexx 3.0" available from
  - http://www.NetRexx.org

NetRexx, 1

- Principles and syntax derived from Rexx
  - Most of the functionality of the classic Rexx BIFs ("built-in-functions") implemented in the NetRexx runtime
  - Source code gets translated transparently to Java
    - Possible to have the intermediate Java code saved in files
  - Java source gets compiled into the JVM (Java Virtual Machine) byte code producing genuine Java classes
    - As a result, like Java
      - *"NetRexx programs get compiled once, but run everywhere" !*
      - E.g. NetRexx used to create Google Android apps !

NetRexx, 2

- Homepage "http://www.NetRexx.org"
  - Links to binary and source programs
  - Links to documentation, e.g.
    - Freely available, updated language specification by the "NetRexx father" Mike F. Cowlishaw
  - Links to mailing lists for questions and discussions
  - Links to Facebook and Twitter pages
  - Links to tools and projects, e.g.
    - Eclipse plugin
    - jEdit plugin
    - Android development environment

NetRexx, 3
A Minimal Example (author: M.D. Hughes)

- NetRexx program to fetch current date

```
/* hello.nrx */

say Date()
```

- Java program current date

```
/* date.java */

import java.util.Date;

public class date {

    public static void main(String[] args) {

        System.out.println( new Date() );

    }

}
```

NetRexx and "ooRexx/JVM", 1

- "ooRexx/JVM"
  - Supported NetRexx from "day one" ("BSF4Rexx")
  - NetRexx RexxString's can be transparently exchanged between NetRexx and ooRexx at runtime
  - "samples/NetRexx" contains the NetRexx equivalents to "samples/Java", which contain programs that demonstrate
    - How to use "ooRexx/JVM" to easily execute programs by the ooRexx interpreter
    - How to pass arguments from NetRexx/Java to ooRexx and fetch return values from ooRexx programs
    - How to catch ooRexx runtime errors and inspect the ooRexx exception information in detail

NetRexx and "ooRexx/JVM", 2

- Using NetRexx
  - Much simpler than Java
  - NetRexx programmers very likely to be acquainted with Rexx
  - Possibly acquainted with ooRexx
- Take full advantage of an external Rexx/ooRexx interpreter
  - Exploit what Rexx/ooRexx supplies
  - Use existing Rexx/ooRexx programs
  - Solve problems hardly solvable in Java/NetRexx

Java Invoking a Rexx Script
An Example using exec()

```
import org.apache.bsf.*;   // BSF support
import java.io.*;          // exception handling

public class TestSimpleExec {

  public static void main (String[] args) throws IOException
  {
    try {
      BSFManager mgr  = new BSFManager ();
      BSFEngine  rexx = mgr.loadScriptingEngine("rexx");
      String     rexxCode = "SAY 'Rexx was here!'";

      rexx.exec ("rexx", 0, 0, rexxCode);

    } catch (BSFException e) { e.printStackTrace(); }
  }
}
```

Output:

```
Rexx was here!
```

NetRexx Invoking a Rexx Script
An Example using exec()

```
import org.apache.bsf.

mgr     =BSFManager()
Rexx    =mgr.loadScriptingEngine("rexx")
rexxCode="SAY 'Rexx was here!'"
do
   rexxEngine.exec("rexx",0,0,rexxCode)
catch e=BSFException
   e.printStackTrace
end
```

Output:

```
Rexx was here!
```

org.apache.bsf.BSFEngine
Methods to Execute Script Code

- exec(srcName,lineNo,colNo,script)
  - Executes script, no arguments, no return value
- eval(srcName,lineNo,colNo,script)
  - Evaluates script, no arguments, returns result value
- apply(srcName,lineNo,colNo,script,names, args)
  - Executes script with arguments, returns result value
- call(object,methodName,args)
  - Executes method in object supplying arguments, returns result value
  - Available only in BSF4ooRexx, see RexxEngine docs
  - Easier to use RexxProxy-methods instead

NetRexx Invoking a Rexx Script, 1
Executing Rexx script using apply()

```
        /* 'Hello world!' with NetRexx */
parse source . . src; parse version v; say src"/"v": Hello world!"

        /* 'Hello world!' using a Rexx interpreter */
rexxCode="parse source . . src; parse version v; say src'/'v': Hello world!'"
org.apache.bsf.BSFManager().apply("rexx",src,0,0,rexxCode,null,null)
```

Output:

```
rgf_01_runRexx.nrx/NetRexx 2.05 14 Jan 2005: Hello world!
rgf_01_runRexx.nrx/REXX-ooRexx_4.1.0(MT) 6.03 2 Nov 2010: Hello world!
```

NetRexx Invoking a Rexx Script, 2 Supplying an Argument to Rexx

```
parse source . . src

rexxCode="parse arg str; say 'Rexx received:' str"

vArgs=Vector()
vArgs.addElement("Hello from NetRexx")
org.apache.bsf.BSFManager().apply("rexx",src,0,0,rexxCode,null,vArgs)
```

Output:

```
Rexx received: Hello from NetRexx
```

NetRexx Invoking a Rexx Script, 3 Returning Edited Argument from Rexx

```
parse source . . src

rexxCode="parse arg str; return reverse(str)"

vArgs=Vector()
vArgs.addElement("Hello from NetRexx")
result=org.apache.bsf.BSFManager().apply("rexx",src,0,0,rexxCode,null,vArgs)

say "NetRexx received:" result
```

Output:

```
NetRexx received: xxeRteN morf olleH
```

NetRexx Invoking a Rexx Script, 4 Querying Process' Environment

```
parse source . . src

rexxCode="return value(arg(1),,'ENVIRONMENT')"

vArgs=Vector()
vArgs.addElement("PATH")
result=org.apache.bsf.BSFManager().apply("rexx",src,0,0,rexxCode,null,vArgs)
say "PATH:" result
```

Possible Output (Line-breaks by Presentation Program):

```
PATH:
E:\jdk1.6.0_18\bin;E:\rony\dev\bsf\src\bsf4oorexx;D:\Programme\Java\jre6\bin\cl
ient;E:\rony\dev\bsf\src\bsf4oorexx;D:\WINDOWS\system32;D:\WINDOWS;D:\WINDOWS\S
ystem32\Wbem;e:\cygwin\bin;e:\rony\tools;E:\vslick\win;E:\Programme\GNU\GnuPG\p
ub;D:\Programme\Gemeinsame
Dateien\GTK\2.0\bin;D:\WINDOWS\system32;D:\WINDOWS;D:\WINDOWS\System32\Wbem;D:\
Programme\TortoiseSVN\bin;D:\Programme\sK1 Project\UniConvertor-
1.1.5\;D:\Programme\sK1 Project\UniConvertor-
1.1.5\DLLs;D:\Programme\QuickTime\QTSystem\;D:\Programme\ooRexx;D:\Programme\SS
H Communications Security\SSH Secure
Shell;E:\jdk1.6.0_18\jre\bin\client;e:\rony\dev\bsf\src\bsf4oorexx
```

NetRexx Invoking a Rexx Script, 5 Interacting with Process' Environment

```
parse source . . src
envName ="RexxLA"
value="<Rexx Language Association>"
rexxCode='if arg()=1 then return value(arg(1),,"ENVIRONMENT");' -
         'call value arg(1),arg(2),"ENVIRONMENT"'

rexxEngine=org.apache.bsf.BSFManager().loadScriptingEngine("rexx")

vArgs1=Vector()
vArgs1.addElement(envName)
result=rexxEngine.apply(src,0,0,rexxCode,null,vArgs1)
say 'Value of process environment variable "'envName'":' '['result']'

say 'Defining a process environment variable named "'envName'" ...'
vArgs2=Vector()
vArgs2.addElement(envName)
vArgs2.addElement(value)
rexxEngine.apply(src,0,0,rexxCode,null,vArgs2)

result=rexxEngine.apply(src,0,0,rexxCode,null,vArgs1)
say 'Value of process environment variable "'envName'":' '['result']'
```

Output:

```
Value of process environment variable "RexxLA": []
Defining a process environment variable named "RexxLA" ...
Value of process environment variable "RexxLA": [<Rexx Language Association>]
```

ooRexx/JVM Samples

- Numerous examples
  - "samples/Java" showcases
    - Invoking ooRexx from Java, supplying a Java object
      - Fetching Java object via its bean name
      - Fetching Java object as an ooRexx object
    - Sending messages from Java to ooRexx objects
    - Catching ooRexx conditions in Java

Parse Java Object by BeanName, 1

- ooRexx/JVM invokes ooRexx program
  - Java objects as arguments are supplied by their BeanName (a string)
    - unique index value into the BSFRegistry
  - ooRexx program needs to use the public routine bsf.wrap() to create an ooRexx proxy object for the Java object in the BSFRegistry
    - bsf.wrap() is defined in the package "BSF.CLS"
    - BeanName needs to have "<O>" prepended to indicate that the string is a reference to a Java object

Parse Java Object by BeanName, 2
"nrxRunRexx_01.nrx"

```
parse source . . src

rexxCode= "parse arg beanName                          ;" -
          "say 'beanName        :' beanName            ;" -
          "javaObj=bsf.wrap('<O>' || beanName)         ;" - /* Insider-Knowhow */
          "say 'javaObj~class   :' javaObj~class       ;" -
          "say 'javaObj~toString:' javaObj~toString    ;" -
          "::requires BSF.CLS                          ;"

vArgs=Vector()
vArgs.addElement( System.getProperties )

org.apache.bsf.BSFManager().apply("rexx",src,0,0,rexxCode,null,vArgs)
```

Possible Output (Line-breaks by Presentation Program):

```
beanName        : java.util.Properties@9971ad
javaObj~class   : The BSF_REFERENCE class
javaObj~toString: {java.runtime.name=Java(TM) SE Runtime Environment,
sun.boot.library.path=E:\jdk1.6.0_18\jre\bin, java.vm.version=16.0-b13,
java.vm.vendor=Sun Microsystems Inc., java.vendor.url=http://java.sun.com/,
path.separator=;, java.vm.name=Java HotSpot(TM) Client VM,
file.encoding.pkg=sun.io, sun.java.launcher=SUN_STANDARD, user.country=AT,
sun.os.patch.level=Service Pack 3,  ...
```

Use Java Object as an ooRexx Object, 1 *"nrxRunRexx_02.nrx"*

- ooRexx/JVM invokes ooRexx program
  - Java objects as arguments are supplied by their BeanName (a string)
    - unique index value into the BSFRegistry
  - If ooRexx/JVM detects that the public routine bsf.wrap() is available in the interpreter instance, then it will create an ooRexx proxy object for each Java object argument, then
    - USE ARG will fetch directly the ooRexx proxy object
    - PARSE ARG would parse the BeanName

Use Java Object as an ooRexx Object, 2

"nrxRunRexx_02.nrx"

```
parse source . . src

rexxEngine=org.apache.bsf.BSFManager().loadScriptingEngine("rexx")
rexxEngine.apply(src,0,0,"::requires BSF.CLS",null,null)

rexxCode= "use arg javaObj                              ;" -
          "say 'javaObj~class   :' javaObj~class        ;" -
          "say 'javaObj~toString:' javaObj~toString     ;" -
          "::requires BSF.CLS                           ;"

vArgs=Vector()
vArgs.addElement( System.getProperties )

rexxEngine.apply(src,0,0,rexxCode,null,vArgs)
```

Possible Output (Line-breaks by Presentation Program):

```
bjavaObj~class   : The BSF_REFERENCE class
javaObj~toString: {java.runtime.name=Java(TM) SE Runtime Environment,
sun.boot.library.path=E:\jdk1.6.0_18\jre\bin, java.vm.version=16.0-b13,
java.vm.vendor=Sun Microsystems Inc., java.vendor.url=http://java.sun.com/,
path.separator=;, java.vm.name=Java HotSpot(TM) Client VM,
file.encoding.pkg=sun.io, sun.java.launcher=SUN_STANDARD, user.country=AT,  ...
```

Sending Messages to ooRexx Objects, 1 *"nrxRunRexx_03.nrx"*

- ooRexx objects can be made available to Java/NetRexx
  - org.rexxla.bsf.engines.rexx.RexxProxy class
  - Supplies methods to send messages to the ooRexx objects via the Java RexxProxy objects
    - Modelled after ooRexx 4 kernel APIs e.g.
      - sendMessage0(msgName)
      - sendMessage1(msgName,arg1)
      - sendMessage2(msgName,arg1,arg2) ...
    - Hence easy for Java/NetRexx programmers to send ooRexx objects messages

Sending Messages to ooRexx Objects, 2

"nrxRunRexx_03.nrx"

```
import org.rexxla.bsf.engines.rexx.
parse source . . src

rexxEngine=org.apache.bsf.BSFManager().loadScriptingEngine("rexx")

rexxCode="d=.directory~new                                                ;" -
         "d~ooRexx='Open Object Rexx'                                     ;" -
         "d~BSF4ooRexx='Bean Scripting Framework for Open Object Rexx'    ;" -
         "return d                                                        ;" -
         "::requires BSF.CLS                                              ;"

rp=RexxProxy rexxEngine.apply(src,0,0,rexxCode,null,null);

say 'rp~ooRexx              = ['rp.sendMessage0("ooRexx")']'
say 'rp~entry("ooRexx")     = ['rp.sendMessage1("entry", "ooRexx")']'
say 'rp["OOREXX"]           = ['rp.sendMessage1("[]",    "OOREXX")']\n'

say 'rp~BSF4ooRexx          = ['rp.sendMessage0("Bsf4ooRexx")']'
say 'rp~entry("BSF4ooRexx") = ['rp.sendMessage1("entry", "BSF4ooRexx")"]"
say 'rp["BSF4OOREXX"]       = ['rp.sendMessage1("[]", "BSF4OOREXX")']'
```

Output:

```
rp~ooRexx              = [Open Object Rexx]
rp~entry("ooRexx")     = [Open Object Rexx]
rp["OOREXX"]           = [Open Object Rexx]

rp~BSF4ooRexx          = [Bean Scripting Framework for Open Object Rexx]
rp~entry("BSF4ooRexx") = [Bean Scripting Framework for Open Object Rexx]
rp["BSF4OOREXX"]       = [Bean Scripting Framework for Open Object Rexx]
```

Catching ooRexx Conditions, 1 *"nrxRunRexx_04.nrx"*

- ooRexx conditions (exceptions) can be caught !
- Can be raised as a result of
  - Running a Rexx program that raises a condition
  - Sending a message to an ooRexx object
- Specific Java exception class
  - org.rexxla.bsf.engines.rexx.RexxException class
  - Method getRexxConditionObject() returns the RexxProxy condition object
    - Java/NetRexx can send ooRexx messages to it

Catching ooRexx Conditions, 2

```
import org.rexxla.bsf.engines.rexx.
parse source . . src

rexxEngine=org.apache.bsf.BSFManager().loadScriptingEngine("rexx")

rexxCode= "a=1                                              \n" -
          "b=0                                              \n" -
          "say a/b  -- cause a syntax error at runtime      \n" -
          "::requires BSF.CLS                                 "
do
    result=rexxEngine.apply(src,0,0,rexxCode,null,null)
catch re=RexxException
    System.out.println("Rexx program threw an exception:\n")
    rp=re.getRexxConditionObject()

    say '    rp.sendMessage0("condition")       : ['rp.sendMessage0("condition")']'
    say '    rp.sendMessage0("code")            : ['rp.sendMessage0("code")']'
    say '    rp.sendMessage0("message")         : ['rp.sendMessage0("message")']'
    say '    rp.sendMessage0("program")         : ['rp.sendMessage0("program")']\n'

    traceBack=RexxProxy rp.sendMessage0("traceback")
   say '    traceBack.sendMessage0("firstItem"):\n\t['traceBack.sendMessage0("firstItem")']'
end
```

Output:

```
Rexx program threw an exception:

    rp.sendMessage0("condition")       : [SYNTAX]
    rp.sendMessage0("code")            : [42.3]
    rp.sendMessage0("message")         : [Arithmetic overflow; divisor must not be zero]
    rp.sendMessage0("program")         : [nrxRunRexx_04.nrx]

    traceBack.sendMessage0("firstItem"):
        [     3 *-* say a/b  -- cause a syntax error at runtime ]
```

Roundup

- NetRexx simplifies creating Java classes and Java apps
  - Rexx-like syntax, hence easy to learn and to use
  - Produces Java source code that can be saved
  - Produces Java classes (JVM byte code)
- ooRexx/JVM
  - Powerful combination of a static (NetRexx/Java) and a dynamic (ooRexx) language
  - Both employ the "human-centric" Rexx-language principles
  - Creates an unmatched "win-win" infrastructure